Добрый день! Есть Табличная часть документа. На колонку ИнвентарьИХозяйственныеПринадлежностиПартияМатериаловВЭксплу­атации установлено событие НачалоВыбора. Для выбора требуется вывести только определенные документы. Как установить отбор по этим документам?Конфигурация Бухгалтерия 3.0

```1c
    &НаКлиенте
    Процедура ИнвентарьИХозяйственныеПринадлежностиПартияМатериаловВЭксплу­атацииНачалоВыбора(Элемент, ДанныеВыбора, СтандартнаяОбработка)

    СтандартнаяОбработка = Ложь;
    ТекущиеДанные = Элементы.ИнвентарьИХозяйственныеПринадлежности.ТекущиеДанные;

    ДанныеСтрокиТаблицы = Новый Структура("Номенклатура");
    ЗаполнитьЗначенияСвойств(ДанныеСтрокиТаблицы, ТекущиеДанные);

    ДанныеОбъекта = Новый Структура("Дата, Организация, ОтправляетсяФизЛицо");
    ЗаполнитьЗначенияСвойств(ДанныеОбъекта, Объект);

    СписокДокументовПартия = СписокДокументовПартияНачалоВыбораНаСервере(ДанныеСтрокиТаблицы, ДанныеОбъекта);

    ФормаВыбора = Документы.ПартияМатериаловВЭксплуатации.ПолучитьФормуВыбора("ФормаВыбора", Элемент, );
    ФормаВыбора.ПараметрТекущаяСтрока = Элемент.Значение;

    ФормаВыбора.ДокументСписок.Отбор.Ссылка.ВидСравнения = ВидСравнения.ВСписке;
    ФормаВыбора.ДокументСписок.Отбор.Ссылка.Использование = Истина;
    ФормаВыбора.ДокументСписок.Отбор.Ссылка.Значение = СписокДокументовПартия;

    ФормаВыбора.Открыть();

    КонецПроцедуры

    &НаСервереБезКонтекста
    Функция СписокДокументовПартияНачалоВыбораНаСервере(ДанныеСтрокиТаблицы, ДанныеОбъекта)

    ОграничениеСубконто = Новый Массив();
    ОграничениеСубконто.Добавить(ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Номенклатура);
    ОграничениеСубконто.Добавить(ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.ПартииМатериаловВЭксплуатации);
    ОграничениеСубконто.Добавить(ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.РаботникиОрганизаций);

    ЗапросПартия = Новый Запрос();
    ЗапросПартия.УстановитьПараметр("Период", ДанныеОбъекта.Дата);
    ЗапросПартия.УстановитьПараметр("Организация", ДанныеОбъекта.Организация);
    ЗапросПартия.УстановитьПараметр("СчетУчетаИнвентаряИХозяйственныхПринадлежностейЗабалансовый­БУ", ПланыСчетов.Хозрасчетный.НайтиПоКоду("МЦ.04"));
    ЗапросПартия.УстановитьПараметр("ОграничениеСубконто", ОграничениеСубконто);

    ЗапросПартия.УстановитьПараметр("Субконто3", ДанныеОбъекта.ОтправляетсяФизЛицо);
    ЗапросПартия.УстановитьПараметр("Субконто1", ДанныеСтрокиТаблицы.Номенклатура);

    ЗапросПартия.Текст =
    "ВЫБРАТЬ
    | ХозрасчетныйОстаткиЗБ.Субконто2 КАК Субконто2
    |ИЗ
    | РегистрБухгалтерии.Хозрасчетный.Остатки(&Период, Счет В ИЕРАРХИИ (&СчетУчетаИнвентаряИХозяйственныхПринадлежностейЗабалансовый­БУ), &ОграничениеСубконто, Организация = &Организация) КАК ХозрасчетныйОстаткиЗБ
    |ГДЕ
    | ХозрасчетныйОстаткиЗБ.Субконто3 = &Субконто3
    | И ХозрасчетныйОстаткиЗБ.Субконто1 = &Субконто1
    | И ХозрасчетныйОстаткиЗБ.Субконто2.Ссылка = (ВЫРАЗИТЬ(ХозрасчетныйОстаткиЗБ.Субконто2.Ссылка КАК Документ.ПартияМатериаловВЭксплуатации))";
    РезультатЗапросаПартия = ЗапросПартия.Выполнить().Выбрать();

    СписокДокументовПартия = Новый СписокЗначений;
    Пока РезультатЗапросаПартия.Следующий() Цикл
    СписокДокументовПартия.Добавить(РезультатЗапросаПартия.Субконто2);
    КонецЦикла;

    Возврат СписокДокументовПартия;

    КонецФункции
```
---
Вот как получилось)

```1c
    ПараметрыФормы = Новый Структура("Отбор", Новый Структура("Ссылка", СписокДокументовПартия));
    ОткрытьФорму("Документ.ПартияМатериаловВЭксплуатации.ФормаВыбора", ПараметрыФормы, Элемент);
```
---
```1c
    ФормаВыбора = ПолучитьФорму("Документ.ПередачаМатериаловВЭксплуатацию.ФормаВыбора");

    ЭлементОтбора = ФормаВыбора.Список.Отбор.Элементы.Добавить(Тип("ЭлементОтбораКомпоновкиДанных"));
    ЭлементОтбора.ЛевоеЗначение = Новый ПолеКомпоновкиДанных("Объект.Ссылка");
    ЭлементОтбора.ВидСравнения = ВидСравненияКомпоновкиДанных.ВСписке;
    ЭлементОтбора.Использование = Истина;
    ЭлементОтбора.ПравоеЗначение = СписокДокументовПередача;
    ЭлементОтбора.РежимОтображения = РежимОтображенияЭлементаНастройкиКомпоновкиДанных.Недоступный;

    ФормаВыбора.ВладелецФормы = ВладелецФормы ;
    ФормаВыбора.Открыть();
```
---
в УФ отбор можно задать через структуру параметров формы.
---
А как с параметрами?

```1c
    ПараметрыОбъекта.Вставить("Ссылка", СписокДокументовПартия);
    ПараметрыФормы = Новый Структура("ПараметрыОбъекта", ПараметрыОбъекта);
    ОткрытьФорму("Документ.ПартияМатериаловВЭксплуатации.ФормаВыбора", ПараметрыФормы, Элемент);
```

Так не получается
---
Ну что-то типа:

```1c
    ЗначениеОтбора = Новый Структура("Номер", "333");
    ПараметрыВыбора = Новый Структура("Отбор", ЗначениеОтбора);

    ОткрытьФорму("Документ.ПриходнаяНакладная.ФормаСписка",ПараметрыВыбора);
```
---
Вот как получилось)

```1c
    ПараметрыФормы = Новый Структура("Отбор", Новый Структура("Ссылка", СписокДокументовПартия));
    ОткрытьФорму("Документ.ПартияМатериаловВЭксплуатации.ФормаВыбора", ПараметрыФормы, Элемент);
```
---
http://infostart.ru/public/565658/